AIIM
Association for Information and Image Management
1100 Wayne Avenue, Suite 1100
Silver Spring, Maryland 20910
301/587-8202

Centimeter

1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 mm

1 2 3 4 5

Inches

1.0 1.1 1.25 1.4 1.6 1.8 2.0 2.2 2.5 2.8 3.2 3.6 4.0 4.5 5.0 5.6 6.3 7.1

MANUFACTURED TO AIIM STANDARDS
BY APPLIED IMAGE, INC.

1 of 1

Conf-9408126--1

SAND94-1326

SPICE Macro Model of a Sprytron with MOSFETs in the Avalanche Mode

Carolyn W. Raney
Component Information and Management Department 2252
Sandia National Laboratories
Albuquerque, NM 87185-1073

Abstract: A SPICE [1] macro model for a triggered vacuum gap, a sprytron, intended for use in rapid discharge circuits such as Exploding Bridge Wire (EBW) applications, is presented. Power MOSFETs in the avalanche mode are utilized as the active switching elements in the model. The macro model is compared for accuracy in predicting the time dependent switching current, switch resistance and voltage drop across the switch using several test circuits. Techniques for extracting model parameters are discussed.

INTRODUCTION

High-power plasma switches are critical components of pulse power devices and electrically activated explosive systems which require high holdoff voltages (1 - 6 kV), large currents (up to 10 kA), and high rates of current rise (> 10^{11} A s^{-1}). The sprytron, a triggered vacuum gap, intended for use in rapid discharge circuits such as Exploding Bridge Wire (EBW) applications, meets these specifications. It is of rugged ceramic/metal construction to withstand hostile environmental conditions and it's vacuum characteristic allows operation in highly radioactive environments. Functionally, they act as normally-open switching elements. This, combined with a low on-state resistance and low forward voltage drop, results in an extremely efficient energy transfer function.

A high power MOSFET when used in the avalanche mode exhibits behavior similar to a sprytron. It is capable of holding off up to 1.2 kV and switching 1 kA in nanoseconds with a very low on-state resistance. For this to occur, the net reactance of the other circuit elements must be sufficiently low (on the order of 0.25 Ω to 0.5 Ω with 100 nH of inductance). Sprytrons typically have a low inductance (10 nH) and a low on-state resistance (10 mΩ - 30 mΩ). Due to their similarity in behavior, the approach presented here is to model the sprytron by utilizing a macro model of MOSFETs in an avalanche condition.

Currently no commercial model for the sprytron exists which is for use with present-day, general purpose, circuit analysis programs like SPICE [1] and PSpice [2]. As it is often both expensive and difficult to simply build and test systems, simulation and accurate models are becoming an important activity. To be able to model a non-semiconductor device, such as the sprytron, utilizing an electronic circuit analysis program like SPICE [1], is not only unique, but affords designers the luxury of implementing design margin analysis on their electrical systems.

The macro model presented in this paper simulates the high power switching characteristics of the sprytron. The techniques for parameter extraction from experimental data will be discussed in detail. A comparison of the bench measurements from several test circuits with the PSpice [2] simulation results will be used to determine the macro model's accuracy in predicting the time dependent switching current, dynamic switch resistance and voltage drop across the switch. The use of this macro model does not require an in-depth understanding of the device characteristics. The macro model is designed to allow the user to easily change sprytron parameters such as trigger probe resistance, capacitance and turn-on voltage.

MACRO MODEL DEVELOPMENT

An example of the drain-source voltage (V_{DS}) and current (I_{DS}) characteristics of a power MOSFET conducting in the avalanche breakdown mode is presented in Figure 1. Figure 2 shows the anode-cathode voltage (V_{AC}) and current (I_{AC}) characteristics of a sprytron. Due to the similarity in the voltage and current characteristic curves it was decided that an avalanche mode MOSFET macro model [3] could be utilized as the basis for a sprytron macro model. A schematic of the sprytron macro model is presented in Figure 3, while a netlist of the macro model is presented in Table 1.



Figure 1. Power MOSFET Conducting in Avalanche Breakdown Mode.



Figure 2. Sprytron Anode-Cathode current and voltage curves.

The sprytron has a region of low current (100 A to 200 A), which I will refer to as a threshold current, which occurs just prior to the avalanche current mode. In order to model these effects, two level-1 MOSFETs are used in parallel. M1 models the threshold current, trigger probe capacitance and resistance, and the trigger turn on voltage while M2 models the avalanche current. LINT represents the internal inductance of the sprytron. RTRIG represents the trigger probe resistance and CTRIG represents the trigger probe capacitance of the sprytron. The sprytron trigger turn on voltage is the parameter VTO in M1. The capacitor CDS is implemented to simulate the anode-cathode voltage fall time. The latch circuit in Figure 3b is used to control the avalanche MOSFET M2. Voltage source VIDSMON is used to control HLATCH, a first-order polynomial current-controlled voltage source. When the voltage across HLATCH reaches approximately 0.7 V, DLATCH begins conducting through R1 and CLATCH. EGSAVAL is a voltage-controlled voltage source whose purpose is to turn MOSFET M2 on. This occurs when the voltage across CLATCH (node 300) reaches 1 volt. M2 has a very low drain resistance, RD_{aval}, which enables M2 to simulate the avalanche current mode of the sprytron.

SPRYTRON MACRO MODEL NETLIST

```
.SUBCKT SPRYTRON      10     20     30     30      400
*                      |      |      |      |      GND
*                      |      |      |  SUBSTRATE
*                      |      |  CATHODE (SOURCE)
*                      |  TRIGGER (GATE)
*               ANODE (DRAIN)
VIDSMON         10A      11        0
RTRIG    20     30       7K
CTRIG    20     30       560PF
LINT            10A      10        4NH
CDS      11     30       .4UF
* CDS is used to simulate the fall time of the anode-cathode voltage.
M1       11     20       30        30        MOSNORM
.MODEL  MOSNORM  NMOS ( LEVEL=1   VTO=98        RD=0.06
** PARAMETERS BELOW INCLUDED FOR DEFAULT VALUES
+ AF=1    CBD=0    CBS=0      CGBO=0  CGDO=0   CGSO=0
+ CJ=0    CJSW=0   DELTA=0    FC=0.5  GAMMA=0  IS=1E-14
+ KF=0    KP=2E-5  LAMBDA=0   LD=0    MJ=0.5   MJSW=0.33
+ NFS=0   PB=0.8   PHI=0.6    RS=0    RSH=0    TPG=1
+ UO=60   VMAX=0   XJ=0       XQC=1   JS=1E-8  UTRA=0
+ NSUB=0  NSS=0
** PSpice (MICROSIM) SPECIFIC PARAMETERS FOLLOW
+ JSSW=0   L=1E-4   N=1      PBSW=0.8   RB=0   RDS=1E12
+ W=1000   RG=0.5   TT=10N   WD=0 )
*The trigger threshold voltage can be varied by changing the parameter
* VTO in mosnorm model parameter set. Be sure that VTO is 2 volts
* less than the trigger voltage, otherwise the MOSFET won't turn on.
M2        11    25     30     30     MOSAVAL
EGSAVAL   25    30     300    400    1
* ---BEGIN LATCH----
*The threshold current can be changed in the HLATCH statement. The
* -1500 value corresponds to a 150A threshold current. Multiply the
*threshold current value times -10 to obtain 1ST coefficient in POLY.
HLATCH   100    400    POLY(1)  VIDSMON  -1500   10
DLATCH   100    200    DLATCH
R1       200    300    1
CLATCH   300    400    10N
RLATCH   300    400    50K
DCLAMP   400    300    DCLAMP
*---END OF LATCH---
.MODEL   MOSAVAL NMOS ( LEVEL=1   VTO=1   RD=0.06
** PARAMETERS BELOW INCLUDED FOR DEFAULT VALUES
+ AF=1    CBD=0    CBS=0      CGBO=0  CGDO=0   CGSO=0
+ CJ=0    CJSW=0   DELTA=0    FC=0.5  GAMMA=0  IS=1E-14
+ KF=0    KP=2E-5  LAMBDA=0   LD=0    MJ=0.5   MJSW=0.33
+ NFS=0   PB=0.8   PHI=0.6    RS=0    RSH=0    TPG=1
+ UO=60   VMAX=0   XJ=0       XQC=1   JS=1E-8  UTRA=0
+ NSUB=0  NSS=0
** PSpice (MICROSIM) SPECIFIC PARAMETERS FOLLOW
+ JSSW=0   L=1E-4   N=1      PBSW=0.8   RB=0   RDS=1E12
+ W=1000   RG=0.5   TT=10N   WD=0 )
.MODEL   DLATCH    D(  IS=1E-14   N=1   BV=40K
**PARAMETERS BELOW INCLUDED FOR DEFAULT VALUES
+ AF=1    CJO=0   EG=1.11   FC=0.5   KF=0   IBV=1E-10
+ M=0.5   RS=0    TT=0      VJ=1     XTI=3
** PSpice (MICROSIM) -SPECIFIC PARAMETERS FOLLOW
+ IBVL=0   IKF=1E12   ISR=0    NBV=1   NBVL=1          NR=2
+ TIKF=0   TBV1=0     TBV2=0   TRS1=0  TRS2=0)
.MODEL   DCLAMP    D ( IS=1E-12     N=1  BV=15
**PARAMETERS BELOW INCLUDED FOR DEFAULT VALUES
+ AF=1    CJO=0   EG=1.11   FC=0.5   KF=0   IBV=1E-10
+ M=0.5   RS=0    TT=0      VJ=1     XTI=3
** PSpice (MICROSIM) -SPECIFIC PARAMETERS FOLLOW
+ IBVL=0   IKF=1E12   ISR=0    NBV=1   NBVL=1          NR=2
+ TIKF=0   TBV1=0     TBV2=0   TRS1=0  TRS2=0)
.ENDS SPRYTRON
```

TABLE 1. NETLIST FOR THE SPRYTRON MACRO MODEL



a) *MOSFET portion of SPRYTRON MACRO MODEL*

b) *Latch circuit controls avalanche mode MOSFET*

Figure 3. Sprytron Macro Model

Parameter Extraction

Figure 4 shows the test circuit used for parameter extraction and model verification. Parameter extraction for the model requires known values of the external circuit elements, measurements from laboratory data (I_{AC} as a function of time), and a series of simple calculations from the characteristic equations [4] for an under damped series RLC circuit. Model parameter values are initially calculated using the method outlined in Table 2. The I_{AC} wave forms provide the timing values needed for the calculation of the total resistance and inductance in the circuit. Once the source values of resistance and inductance have been measured for the external circuit elements, including parasitics from wires, leads, and connections, the sprytron model parameters can be calculated. All but two parameters (VTO and RD) in the MOSFET model statements use the default values.

SPRYTRON MACRO MODEL PARAMETER CALCULATIONS

ω_o = IDEAL RADIAN FREQUENCY =
NATURAL FREQUENCY OF AN L-C CIRCUIT =
$2 * \pi * f = 2 * \pi * (1 / T) = (L * C)^{(-1/2)}$
f = FREQUENCY OF I_{AC} CURRENT WAVE FORM =1 / T
T = PERIOD OF I_{AC} CURRENT WAVE FORM
ω_d = REAL RADIAN FREQUENCY = $(\omega_o^2 - \alpha^2)^{1/2}$
$\alpha = R_{total} / (2 * L_{total})$
R_{total} = EFFECTIVE TOTAL RESISTANCE OF TEST CIRCUIT =
REXT + R_{para} + RCVR
REXT = EXTERNAL LOAD RESISTANCE
(USE R_{total} - R_{para} - RCVR IN SIMULATIONS)
$R_{para} = RD_{norm} \, // \, RD_{aval}$
RD_{norm} = DRAIN RESISTANCE VALUE OF MOSFET M1=
$RD_{aval} = 2 * R_{para}$
RD_{aval} = DRAIN RESISTANCE VALUE OF MOSFET M2 =
$RD_{norm} = 2 * R_{para}$
CEXT=TOTAL EXTERNAL CAPACITANCE IN TEST CIRCUIT
L_{total} = EFFECTIVE TOTAL INDUCTANCE OF TEST CIRCUIT =
LEXT + L_{int}
L_{int} = SPRYTRON MACRO MODEL INDUCTANCE
LEXT = EXTERNAL CIRCUIT INDUCTANCE
(USE L_{total} - L_{int} IN SIMULATIONS)

Calculations for determining:
R_{total}, L_{total}, R_{para}, RD_{aval}, RD_{norm}, α, and ω_o

---

$$i(t) = A e^{-\alpha t} \cos(\omega_d t) + B e^{-\alpha t} \sin(\omega_d t)$$
$$i(0) = A = 0$$
$$\therefore \quad i(t) = B e^{-\alpha t} \sin(\omega_d t)$$
$$di/dt = \omega_d \, B e^{-\alpha t} \cos(\omega_d t) - B e^{-\alpha t} \sin(\omega_d t)$$
Solve for max/min $\Rightarrow di/dt = 0 \Rightarrow$
$$\omega_d = \alpha \, (\sin(\omega_d t) / \cos(\omega_d t))$$
or
$$\alpha = \omega_d \, (\cos(\omega_d t) / \sin(\omega_d t))$$
Use the lab data peaks to solve for several values of α and average them.
Solve for $\omega_o = (\omega_d^2 + \alpha^2)^{1/2}$
$$L_{total} = (1 / (\omega_o^2 * C))$$
$$R_{total} = \alpha * 2 * L_{total}$$
$$R_{para} = R_{total} - REXT$$
$$RD_{aval} = RD_{norm} = 2 * R_{para}$$

Table 2. Details of Sprytron Macro Model Parameter Calculations



Figure 4. Test circuit for parameter extraction and model verification.

Model Evaluation

Actual applications rely on the anode-cathode current supplied to the load. Data was taken using a Tektronix DSA602A digitizing oscilloscope. Although the voltages are displayed in the figures, there was not a requirement to match these. The voltage measurements were extremely difficult to obtain. The 1000X Tektronix 6015A voltage divider probe which was used to take the voltage measurements only had an accuracy of approximately 10%. In addition, the measurements had to be obtained at 100 MHz bandwidth which introduced noise. We obtained measurements at a 20 MHz bandwidth which resulted in less noise, but altered the fall time of the voltage waveform. Therefore, we had to use 100 MHz bandwidth measurements. Figures 5 and 6 display the voltage and current data for 5 shots using the same sprytron and load circuit. The voltage peak values vary as much as 67% (excluding the -1000 V spikes). There is variation in current peak values of as much as 6% on the second and third peaks. The customer's requirements for the macro model were that the first current peak for I_{AC} be within 5 % of the measured value and that the second and third peaks be within 10 %. The timing parameter requirements for the I_{AC} wave form were that the first three peaks and pulse widths be within 8 % of the measured values. The macro model met or exceeded these requirements. We verified the macro model against measured values for several different loads, capacitors, and anode voltages using the circuit in Figure 4.

The sprytron which we measured had a minimum holdoff voltage of 10 kV, a trigger probe resistance of 1 kΩ, a minimum trigger voltage of 100 V, and a trigger capacitance of 650 pF. Measured laboratory data and simulated PSpice [2] data are shown in Figure 7 for test conditions of V_{CC} = 5575 V , CEXT = 0.5 uF, LEXT = 321 nH, REXT = 543 m Ω, and RCVR = 5 mΩ. Figure 8 displays the measured and simulated voltage and current data for a different load condition (VCC = 5575 V, CEXT = 0.06 uF, LEXT = 401 nH, REXT = 269 mΩ and RCVR = 5 mΩ). The macro model accurately simulates the peak values, decay characteristics, and ring down frequency of I_{AC}. The macro model accurately simulates the decay characteristics of V_{AC}, however better measurement techniques need to be explored in order to obtain accurate V_{AC} waveform data.



Figure 5. Measured anode-cathode voltage for 5 shots.



Figure 6. Measured anode-cathode current data for 5 shots.



Figure 7. Anode-Cathode voltage and current characteristics (measured and simulated).



Figure 8. Anode-Cathode voltage and current characteristics (measured and simulated).

SUMMARY and CONCLUSIONS

A macro model of a high energy switch, a sprytron, has been presented. The model is SPICE2G.6 compatible which allows it to be used in the broadest range of circuit simulators. The model accurately simulates the firing characteristics of the anode-cathode current. The model is unique because it utilizes an electronic circuit analysis program to model a non-semiconductor device. The model parameters are relatively easy to extract from measured data. The use of this macro model does not require an in-depth understanding of the device characteristics. The macro model is designed to allow the user to easily change sprytron parameters such as trigger probe resistance, capacitance and turn-on voltage. Future work could be done to improve the V_{AC} phase and amplitude characteristics. However, the model allows a designer to explore variations in circuit loads, thus enabling the designer to optimize his circuit design analytically without having to resort to the more expensive method of laboratory bench testing.

ACKNOWLEDGMENTS

The author wishes to thank Gordon Boettcher, Ronald Brooks, Roger Roberts, Roger Edwards, Alan Vigil, Wallace Newman, Fred Anderson, John Rosborough and Michael Deveney of Sandia National Laboratories for providing information and data for the initial design of the model. This work was supported by the United States Department of Energy under Contract DE-AC04-94AL85000.

REFERENCES

[1] SPICE2G.6, Department of Electrical Engineering and Computer Sciences, University of California.
[2] PSpice Circuit Analysis Version 6.0, MicroSim Corporation, January 1994.
[3] C. Helms, M. Ackermann, T. Fischer, and M. Deveney, "SPICE Macromodel for a 1-Megawatt Power MOSFET Switch," Proceedings of the 36^{th} Midwest Symposium on Circuit and Systems, Vol. 2, 1994.
[4] J. W. Nilsson, "Electric Circuits," pp. 229-234, Addison-Wesley Inc., 1983.

DISCLAIMER

DATE
FILMED
9 /28 /94

END